

**CMPSCI 105 Fall 2009**  
**Quiz #4 Solution Key**  
 © Professor William T. Verts

	A	B	C	D	E	F	G	H	I	J
1										
2			<b>Exam</b>	<b>Grade</b>	<b>Note</b>					
3		<b>Fred</b>	82				0	F	FAIL	
4		<b>Sam</b>	35				60	D	WARNING	
5		<b>Mary</b>	94				70	C	OK	
6		<b>Tom</b>					80	B	GOOD	
7		<b>Carol</b>	78				90	A	PASS	

1. Write a formula for cell **D3** to assign Fred a grade based on his exam score. Write the formula so that it can be copied to cells **D4** through **D7**.

**Answer:**     =VLOOKUP (C3 , \$G\$3 : \$I\$7 , 2)

                  =VLOOKUP (C3 , \$G\$3 : \$H\$7 , 2)

                  =IF (C3 >= \$G\$7 , \$H\$7 ,  
                       IF (C3 >= \$G\$6 , \$H\$6 ,  
                       IF (C3 >= \$G\$5 , \$H\$5 ,  
                       IF (C3 >= \$G\$4 , \$H\$4 , \$H\$3) ) ) )

**Discussion:** The intended answer uses the VLOOKUP function with the given table. Because the formula needs to be copied down for all the rows, the table must be specified using absolute addressing (the dollar signs) to keep the table reference from “crawling” down for each row. The table *must* include columns G and H, and *may* include column I as well (which will be required for question #2). The third parameter of VLOOKUP represents the column of the table containing the return value: the search values are in column 1 of the table, the letter grades are in column 2, and the “notes” are in column 3.

The form =VLOOKUP (C3 , GRADES , 2) *may* be used, but only so long as the students note that the range name GRADES is explicitly assigned to the range G3:H7 or the range G3:I7. (When you define a range name in Excel you automatically get the dollar signs.)

The “brute-force” approach with a four-IF network can be used as well, but the formula *must* use cell addresses from the provided table, and not explicit constants such as 90 and "A". The cell addresses must also be absolute.

**Scoring:**     3 points.

If the student uses the VLOOKUP function, remove 1 point for specifying the range as G3:I7 or G3:H7 instead of \$G\$3:\$I\$7 or \$G\$3:\$H\$7 (forgetting the dollar signs). Since the formula is to be copied down, and not to the left or

right, the mixed addressing forms G\$3:I\$7 or G\$3:H\$7 are allowed. Remove 1 point for using \$C\$3 (absolute addressing) instead of C3. Remove ½ point for specifying the return column (the third parameter) as H instead of 2. Remove ½ point for all other syntax errors, such as forgetting the equal sign, misuse of commas, parentheses, etc. Remove 1 point if the student uses a range name such as GRADES to indicate the table but does not also specify that the range name applies to the range G3:I7 or G3:H7.

If the student uses the four-IF network, remove 1 point if the cell references in columns G and H are relative instead of absolute (forgetting the dollar signs). Remove 1 point for using \$C\$3 (absolute addressing) instead of C3. Remove 2 points if they use explicit constants instead of the provided table, as in:

```
=IF(C3>=90,"A",IF(C3>=80,"B",IF(C3>=70,"C",IF(C3>=60,"D","F"))))
```

Remove ½ point for all other syntax errors, such as forgetting the equal sign, misuse of commas, parentheses, etc. Do not go below zero points.

- 
2. Write a formula for cell E3 to assign Fred a “note” based on his exam score. Write the formula so that it, too, can be copied down.

**Answer:**        =VLOOKUP(C3,\$G\$3:\$I\$7,3)

```
=IF(C3>=$G$7,$I$7,  
    IF(C3>=$G$6,$I$6,  
        IF(C3>=$G$5,$I$5,  
            IF(C3>=$G$4,$I$4,$I$3))))
```

**Discussion:** The discussion for this problem is identical to that for problem #1, with the exception that the return column of VLOOKUP is column 3 of the table, corresponding to physical column I in the brute-force approach. The range for the table must be \$G\$3:\$I\$7, and cannot stop with column H as in problem #1.

**Scoring:** 3 points. The scoring for this problem is identical to that for problem #1, except that VLOOKUP must use column 3 and the range must be \$G\$3:\$I\$7. For the four-IF solution the return values are all from column I.

---

3. Write a formula that computes the date exactly **1000** days after December 31, 2009.

**Answer:**        **=DATE (2009 , 12 , 31) +1000**

**Discussion:** The date is fixed, but is not the current date (November 10<sup>TH</sup>), so we'll need to compute a serial date number manually with the DATE function. The order of the parameters is always largest-unit-of-time down to smallest-unit-of-time, which is year, then month, then day. Once we have the serial date number we add 1000 to that number to project the date into the future. Putting the 1000 in the wrong place, such as =DATE (2009, 12, 31+1000) actually works correctly, and is allowed (although it is not the preferred form).

**Scoring:** 2 points. Remove ½ point for using the wrong function name, such as NOW(2009, 12, 31). Remove ½ point for specifying the parameters in the wrong order, such as DATE (12, 31, 2009). Remove ½ point for all syntax errors, such as forgetting the equal sign, misuse of commas, parentheses, etc. Remove ½ point for any errors not covered here.

---

4. What time of day is it if the =NOW ( ) function returns **40127.625** exactly?

**Answer:**        3:00pm exactly (or 15:00 in military format).

**Discussion:** The 40127 represents the date (corresponding to November 10, 2009), not the time, and thus is not part of this problem. The fraction 0.625 represents the fraction of the day since midnight, and that is what we need to solve the problem. Some people might recognize 0.625 as equivalent to the fraction  $\frac{5}{8}$ , and since there are 24 hours in a day,  $24 \times \frac{5}{8} = 15$ . Similarly, you can do the multiplication directly to compute  $24 \times 0.625 = 15$ . The time is then 15 hours since midnight, or 3:00pm exactly.

**Scoring:** 2 points. Accept either 3:00pm or 15:00. Do not accept “early afternoon” or “mid afternoon” as the answer. Remove 1 point for “around 3:00pm” but do not accept any other vague time.

---